Using the PITCH Instruction

Real time pitch transposing (often incorrectly called pitch shifting) is available in the FXCore by use of a ramp wave generator and the PITCH instruction.

Normally in FXCore the address to access delay memory in an instruction is added to a down counter that is decremented once per sample period, this causes the entire memory to operate like a large circular delay element.

As we are using a down counter to create the circular memory we must therefore use a down ramp to increase the rate we read from the delay memory to create a pitch up effect and an up ramp to slow down the rate to create a pitch down effect.

To pitch up we set the coefficient to the ramp as:

$C = -2^{23} * (2^N - 1)*(512/L)$

N = Desired amount of pitch shift in octaves
L = Length of delay line used (512, 1024, 2048 or 4096)

For pitching down it is:

$C = 2^{23} * (1-(1/2^{N)}))*(512/L)$

N = Desired amount of pitch shift in octaves
L = Length of delay line used (512, 1024, 2048 or 4096)

While the pitch instruction its self is very simple, generating the coefficients can be a bit more complex. For example assume we want to write a pitch program that uses POT0 to adjust the pitch +/-1 octave, the pitch should be -1 octave when POT0 is fully counter-clockwise, 0 when POT0 is centered and +1 octave when POT0 is fully clockwise.

Using a delay length of 4096 samples we calculate a ramp coefficient of -1048576 (0xFFF0000) for +1 octave and 524288 (0x00080000) for -1 coefficient, we now need to make POT0 sweep between this range calculating the correct coefficient for the ramp and setting the ramp control register to this value.

We can do this in a few FXCore instructions if we consider a few things:

POT0 ranges from 0 to 0.99… which we will consider close enough to 1.0 for our use

The +1 octave coefficient is 2 times the -1 octave coefficient and the opposite sign

We want a coefficient of 0 when POT0 is centered, so if we subtract 0.5 from the POT0 value it will now range -0.5 to +0.5

We actually want a positive value when we are fully counter-clockwise so we multiply the adjusted POT0 value by -1048576 (0xFFF00000) since: -0.5*-1048576 = 524288

If the result of the multiplication is negative it means we are above the midpoint and actually want to shift up and as the up coefficient is 2 times the size of the down coefficient we simply multiply the result by 2.

```
// Example using the PITCH instruction - an-2.fxc
// This example adjusts the pitch +/-1 octave using
// POT0 to control the amount of shift

// Define the delay block for the pitch delay
.mem pdelay 4096

// Read in the POT0 value to R0, ranges 0 to 1.0(almost)
cpy_cs r0, pot0_smth

// Put -0.5 into R1, 0xC000 goes into R1[31:16] and 0s into R1[15:0]
wrdld r1, 0xC000

// Adding them makes it range -0.5 to +0.5
adds  r0, r1

// Put -1048576 into R0, 0xFFF0 goes into R0[31:16] and 0s into R0[15:0]
wrdld r0, 0xFFF0

// Multiply the adjusted POT0 value by -1048576
multrr acc32, r0

// If positive jump over the multiply by 2
jgez acc32, OK

// Do the multiply by shifting left 1 bit
sls acc32, 1

// Write the result to the ramp0 frequency control
OK: cpy_sc ramp0_f, acc32

// Read channel 0 input
cpy_cs acc32, in0

// Write it to the delay
wrdel pdelay, acc32

// Do the shift, result will be in ACC32
pitch rmp0|14096, pdelay

// Write to output
cpy_sc out0, acc32
```

Experimental Noize Inc. reserves the right to make changes to, or to discontinue availability of, any product or service without notice.

Experimental Noize Inc. assumes no liability for applications assistance or customer product design. Customers are responsible for their products and applications using any Experimental Noize Inc. product or service. To minimize the risks associated with customer products or applications, customers should provide adequate design and operating safeguards.

Experimental Noize Inc. make no warranty, expressed or implied, of the fitness of any product or service for any particular application.

In no even shall Experimental Noize Inc. be liable for any direct, indirect, consequential, punitive, special or incidental damages including, without limitation, damages for loss and profits, business interruption, or loss of information arising out of the use or inability to use any product or document, even if Experimental Noize Inc. has been advised of the possibility of such damage.

SAFETY-CRITICAL, MILITARY, AND AUTOMOTIVE APPLICATIONS DISCLAIMER: Experimental Noize Inc. products are not designed for and will not be used in connection with any applications where the failure of such products would reasonably be expected to result in significant personal injury or death (“Safety-Critical Applications”). Safety-Critical Applications include, without limitation, life support devices and systems, equipment or systems for the operation of nuclear facilities and weapons systems. Experimental Noize Inc. products are not designed nor intended for use in military or aerospace applications or environments. Experimental Noize Inc. products are not designed nor intended for use in automotive applications.

Experimental Noize Inc.

Scottsdale, AZ USA

www.xnoize.com

sales@xnoize.com